Bridging the Gap between TDPE and SDPE

Eijiro Sumii
Department of Information Science, Graduate School of Science, University of Tokyo
Roadmap

(1) Naive Online SDPE

(2) Offline TDPE

(4) Cogen Approach to Online SDPE

(3) Online TDPE
Roadmap

(1) Naive Online SDPE

(2) Offline TDPE

(4) Cogen Approach to Online SDPE

(3) Online TDPE
What is Partial Evaluation?

• Partial Evaluation = Program Specialization

\[ p = \lambda s. \lambda d. 1 + s + d \]

\[ \downarrow s = 2 \]

\[ p_2 = \lambda d. 3 + d \]

• Partial Evaluation \approx Strong Normalization

\[ (\lambda s. \lambda d. 1 + s + d) \ @ 2 \]

\[ \rightarrow \lambda d. 1 + 2 + d \]

\[ \rightarrow \lambda d. 3 + d \]
Naive Syntax-Directed PE

- Represent Programs as Data
  \[ e ::= x \mid \lambda x. e \mid e_1 \circ e_2 \]
- Manipulate Them Symbolically
  \[
  \begin{align*}
  PE(x) &= x \\
  PE(\lambda x. e) &= \lambda y. PE(e[y/x]) \quad \text{(where } y \text{ is fresh)} \\
  PE(e_1 \circ e_2) &= PE(e_1[PE(e_2)/x]) \quad \text{(if } PE(e_1) = \lambda x. e) \\
  PE(e_1 \circ e_2) &= PE(e_1) \circ PE(e_2) \quad \text{(otherwise)}
  \end{align*}
  \]
Implementation in ML

datatype exp = Var of string
  | Abs of string * exp
  | App of exp * exp

... 

fun PE (Var(x)) = Var(x)
| PE (Abs(x, e)) = 
  let val y = gensym ()
  in Abs(y, PE (subst x (Var(y)) e))
  end
| PE (App(e1, e2)) = 
  let val e1' = PE e1
      val e2' = PE e2
  in (case e1' of
       Abs(x, e) => PE (subst x e2' e)
       | e => App(e, e2'))
  end
Example

Partially Evaluate $p = \lambda s. \lambda d. s @ d$

with Respect to $s = \lambda x. x$

$\approx$ Strongly Normalize

$p @ s = (\lambda s. \lambda d. s @ d) @ (\lambda x. x)$

```
- let val p = Abs("s",
    Abs("d",
        App(Var "s",
            Var "d")))

    val s = Abs("x", Var("x"))

    in PE (App(p, s))
end;
> val it = Abs ("x1",Var "x1") : exp
```
Problems of Naive SDPE

• Naive SDPE is Complex
  – Includes an Interpreter
  – Requires one clause in the partial evaluator for one construct in the target language

• Naive SDPE is Inefficient
  – Incurs interpretive overheads such as:
    • syntax dispatch
    • environment manipulation
Roadmap

(1) Naive Online SDPE
(2) Offline TDPE
(3) Online TDPE
(4) Cogen Approach to Online SDPE
Type-Directed PE [Danvy 96]

• Originates in *Normalization by Evaluation* in Logic and Category Theory

\[
\begin{align*}
&\text{value} \\
&\text{eval} \uparrow \downarrow \text{reify} \\
&\text{exp} \\
&\text{normalize} = \text{reify} \circ \text{eval}
\end{align*}
\]

• Exploit the Evaluator of the Meta Language
Example

```plaintext
- let fun p s d = s d
  fun id x = x
  val p_id = p id
  in reify (E-->E) p_id
  end;

> val it = Abs ("x1",Var "x1") : exp
```
How to Reify?
— When the Domain is a Base Type —

• $\downarrow_{\alpha \rightarrow \alpha} v = \lambda x. v \ @ x$
  
  e.g. $\downarrow_{\alpha \rightarrow \alpha} (\lambda x. (\lambda y. y) \ @ x)$
  
  $= \lambda z. (\lambda x. (\lambda y. y) \ @ x) \ @ z$
  
  $= \lambda z. (\lambda y. y) \ @ z$
  
  $\__ = \lambda z. z$

• $\downarrow_{\alpha \rightarrow \tau} v = \lambda x. \downarrow_\tau (v \ @ x)$
  
  e.g. $\downarrow_{\alpha \rightarrow \alpha \rightarrow \alpha} (\lambda x. \lambda y. x)$
  
  $= \lambda p. \downarrow_{\alpha \rightarrow \alpha} ((\lambda x. \lambda y. x) \ @ p)$
  
  $= \lambda p. \lambda q. (\lambda x. \lambda y. x) \ @ p \ @ q$
  
  $\__ = \lambda p. \lambda q. q$
In ML...

- let val f = fn x => (fn y => y) x
  val z = gensym ()
  in Abs(z, f (Var(z)))
  end;
> val it = Abs ("x1",Var "x1") : exp
- let val g = fn x => fn y => x
  val p = gensym ()
  val q = gensym ()
  in Abs(p, Abs(q, g (Var(p)) (Var(q))))
  end;
> val it = Abs ("x2",Abs ("x3",Var "x2")) : exp
How to Reify?
— When the Domain is a Function Type —

\[ \downarrow_{(\alpha \to \alpha) \to \alpha} v = \lambda x. \ v \ @ \ x \quad \Leftarrow \text{Type Error} \]

\[ \downarrow_{(\alpha \to \alpha) \to \alpha} v = \lambda x. \ v \ @ \ (\lambda y. \ x \ @ \ y) \]

e.g. \[ \downarrow_{(b \to b) \to b} (\lambda f. \ f \ @ \ x) \]
\[ = \lambda y. \ (\lambda f. \ f \ @ \ x) \ @ \ (\lambda z. \ y \ @ \ z) \]
\[ = \lambda y. \ (\lambda z. \ y \ @ \ z) \ @ \ x \]
\[ = \lambda y. \ y \ @ \ x \]
In ML...

- let val h = fn f => f (Var("x"))
  
  val y = gensym ()
  
  in Abs(y, h (Var(y)))
  
  end;

  Error: operator and operand don't agree
  
  operator domain: exp -> 'Z
  
  operand: exp
  
  in expression: h (Var y)

  - let val h = fn f => f (Var("x"))
    
    val y = gensym ()
    
    in Abs(y, h (fn z => App(Var(y), z)))
    
    end;

  > val it = Abs ("x1",App (Var "x1",Var "x")) : exp
How to Reify?
— In General —

\[ \downarrow : [\tau] \rightarrow \tau \rightarrow \text{exp} \]

\[ \downarrow_\alpha v = v \]

\[ \downarrow_{\sigma \rightarrow \tau} v = \lambda x. \downarrow_\tau (v \@ \uparrow_\sigma x) \]

(\text{where } x \text{ is fresh})

\[ \uparrow : [\tau] \rightarrow \text{exp} \rightarrow \tau \]

\[ \uparrow_\alpha e = e \]

\[ \uparrow_{\sigma \rightarrow \tau} e = \lambda x. \uparrow_\tau (e \@ \downarrow_\sigma x) \]
Implementation in ML (1)

• Straightforward Implementation Fails to Type-Check, Because \(\downarrow\) and \(\uparrow\) are Dependent Functions

• Solution: Represent a Type \(\tau\) by a Pair of Functions \((\downarrow_{\tau}, \uparrow_{\tau})\) [Yang 98] [Rhiger 99]

```ml
datatype 'a typ = RR of ('a -> exp) * (exp -> 'a)

(* reify : 'a typ -> 'a -> exp *)
fun reify (RR(f, _)) v = f v

(* reflect : 'a typ -> exp -> 'a *)
fun reflect (RR(_, f)) e = f e
```
Implementation in ML (2)

(* E : exp typ *)
val E = RR(fn v => v, fn e => e)

(* --> : 'a typ * 'b typ -> ('a -> 'b) typ *)
infixr -->
fun (dom --> codom) = 
RR(fn v =>
    let val x = gensym ()
    in Abs(x, reify codom (v (reflect dom (Var(x)))))
    end,
  fn e =>
  fn x => reflect codom (App(e, reify dom x))
)
Example

```plaintext
  let val S = fn f =>
    fn g =>
      fn x => (f x) (g x)
  in
    val K = fn a => fn b => a
    val I = S K K
    in reify (E-->E) I
  end;

  > val it = Abs ("x1",Var "x1") : exp
```
More Examples

- We can use constructs of ML (but cannot residualize them)
  - `reify (E→E→E)`
    
    
    `(fn x => fn y => if 3+5<7 then x else y);`

    > val it = Abs ("x1",Abs ("x2",Var "x2")) : exp

- We may specify a non-principal type (but get a redundant result)
  - `reify ((E→E)→(E→E)) (fn x => x);`

    > val it = Abs ("x3",Abs ("x4",
                             App (Var "x3",Var "x4"))) : exp
Extensions (1): Pair Types

\[ e ::= \ldots \mid \text{pair}(e_1, e_2) \mid \text{fst} \ e \mid \text{snd} \ e \]

\[ \downarrow_{\sigma \times \tau} \ v = \text{pair} (\downarrow_{\sigma} \ \text{fst} \ v, \downarrow_{\tau} \ \text{snd} \ v) \]
\[ \uparrow_{\sigma \times \tau} \ e = \text{pair} (\uparrow_{\sigma} \ \text{fst} \ e, \uparrow_{\tau} \ \text{snd} \ e) \]
Extensions (2): Variant Types

\[ e ::= \ldots \mid \text{true} \mid \text{false} \mid \text{if } e_0 \text{ then } e_1 \text{ else } e_2 \]

\[ \downarrow_{\text{bool}} v = \text{if } v \text{ then } \text{true} \text{ else } \text{false} \]

\[ \uparrow_{\text{bool}} e = \text{???} \]

— Want to return both "true" and "false" to the context and use the results

⇒ Manipulate *partial continuation* with "shift" & "reset" [Danvy & Finlinski 90]
Problems

• Reflection for variant types causes code duplication

\[ \downarrow_{(\alpha \to \alpha) \to \text{bool} \to \alpha} (\lambda f. \lambda x. f \circ (f \circ (f \circ (\text{if } x \text{ then } y \text{ else } z)))) \]

\[ = \lambda f. \lambda x. \text{if } x \text{ then } f \circ (f \circ (f \circ (y))) \text{ else } f \circ (f \circ (f \circ (z))) \]

• Reflection for primitive/inductive types is impossible

\[ \downarrow_{\text{int} \to \text{int}} (\lambda n. 1 + 2 + n) = ??? \]

\[ \downarrow_{\text{int} \_\text{list} \to \text{int} \_\text{list}} (\lambda a. (\text{tl} (\text{tl} (3 :: a)))) = ??? \]
Roadmap

(1) Naive Online SDPE
(2) Offline TDPE
(3) Online TDPE
(4) Cogen Approach to Online SDPE
Online TDPE (1)

- Extend some primitive operators to treat residual code [Danvy 97]
  \[x +' y = x + y\] (if \(x\) and \(y\) are integers)
  \[x +' y = \downarrow_{\text{int}} x \pm \downarrow_{\text{int}} y\] (if \(x\) or \(y\) is residual code)

For example:
\[
\downarrow_{\text{int} \rightarrow \text{int}} (\lambda n. 1 +' 2 +' n)
= \lambda x. (\lambda n. 1 +' 2 +' n) \ @ \ x \\
= \lambda x. 1 +' 2 +' x \\
= \lambda x. 3 +' x \\
= \lambda x. 3 + x
\]
In ML...

datatype 'a tlv = S of 'a | D of exp

val I = (* int tlv typ *)

RR(fn D(e) => e |
    | S(i) => Int(i),
    fn e => D(e))

fun add' (S(i), S(j)) = S(i + j) |
    add' (x, y) = D(Add(reify I x, reify I y))

- reify (I-->I)
  (fn n => add'(add'(S(1), S(2)), n));

> val it = Abs ("x1",Add (Int 3,Var "x1")) : exp
Online TDPE (2)

• Extend any value destructors to treat residual code [Sumii & Kobayashi 99]

\[ tl' \ x = tl \ x \quad \text{(if } x \text{ is a list)} \]
\[ tl' \ x = \text{tl} \ x \quad \text{(if } x \text{ is residual code)} \]

For example:
\[
\downarrow_{\text{int_list} \rightarrow \text{int_list}} \ (\lambda a. \ (tl' \ (tl' \ (3 :: a))))
= \lambda x. \ (\lambda a. \ (tl' \ (tl' \ (3 :: a)))) \ @ \ x
= \lambda x. \ tl' \ (tl' \ (3 :: x))
= \lambda x. \ tl' \ x
= \lambda x. \ tl \ x
\]
In ML...

datatype 'a list = nil | :: of 'a * 'a list tlv
fun L t = (* 'a typ -> 'a list tlv typ *)
  RR(fn D(e) => e
      | S(nil) => Nil
      | S(x :: y) => Cons(reify t x,
                       reify (L t) y),
       fn e => D(e))
fun tl' (D(e)) = D(Tl(e))
  | tl' (S(_ :: x)) = x

- reify (L I --> L I)
  (fn a => tl' (tl' (S(S(3) :: a)))));
> val it = Abs ("xl",Tl (Var "xl"): exp
Online TDPE (3)

- Extend all value destructors to treat residual code [Sumii & Kobayashi 99]

\[
f @' x:\tau = f @ x \quad \text{(if f is a function)}
\]
\[
f @' x:\tau = f @ \downarrow_\tau x \quad \text{(if f is residual code)}
\]

\[\Rightarrow\text{Reflection becomes unnecessary!}\]
An Experiment

Specialized & executed an interpreter for a simple imperative language with a tiny program (by SML/NJ 110.0.3 on UltraSPARC 168 MHz with 1.2 GB Main Memory)

<table>
<thead>
<tr>
<th></th>
<th>spec</th>
<th>exec</th>
</tr>
</thead>
<tbody>
<tr>
<td>(No PE)</td>
<td>0.30</td>
<td></td>
</tr>
<tr>
<td><a href="*">Danvy 96</a>1</td>
<td>0.57</td>
<td>0.14</td>
</tr>
<tr>
<td><a href="*">Danvy 97</a>2</td>
<td>0.24</td>
<td>0.13</td>
</tr>
<tr>
<td><a href="*">Sumii 99</a>2</td>
<td>0.10</td>
<td>0.14 (msec)</td>
</tr>
</tbody>
</table>

(*1) abstracted out all primitive operators

(*2) removed unnecessary 's by monovariant BTA
Roadmap

(1) Naive Online SDPE

(2) Offline TDPE

(3) Online TDPE

(4) Cogen Approach to Online SDPE
Cogen Approach to Online SDPE

We are going to:

• Realize a simple & efficient online SDPE by using:
  – Higher-Order Abstract Syntax
  – Deforestation

• See a similarity between the SDPE and our online TDPE
Higher-Order Abstract Syntax

Represent binding in the target language by binding in the meta language

```
datatype hexp = HAbs of hexp -> hexp
  | HApp of hexp * hexp
  | HSym of string
```

For example,

```
HAbs(fn x => HApp(HAbs(fn y => y), x)) : hexp
```

represents \( \lambda x. (\lambda y. y) \at x \)
Converter from HOAS to FOAS

fun conv (HAbs(f)) = 
    let val x = gensym ()
    in Abs(x, conv (f (HSym(x))))
    end
| conv (HApp(e1, e2)) = App(conv e1, conv e2)
| conv (HSym(s)) = Var(s)
Online SDPE in HOAS

fun PE (HAbs(f)) = HAbs(fn x => PE (f x))
| PE (HApp(e1, e2)) = 
  let val e1' = PE e1
  val e2' = PE e2
  in (case e1' of HAbs(f) => f e2'
           | _ => HApp(e1', e2'))
  end
| PE (HSym(s)) = HSym(s)

> let val e = HAbs(fn x =>
    HApp(HAbs(fn y => y),
         x))
  in conv (PE e)
  end;
> val it = Abs ("x1",Var "x1") : exp
A priori compose \texttt{HAbs}, \texttt{HApp} & \texttt{HSym} with \texttt{PE}

(Instead of first constructing an \texttt{hexp} by \texttt{HAbs}, \texttt{HApp} & \texttt{HSym} and then destructing it by \texttt{PE})

\begin{verbatim}
fun habs'(f) = HAbs(fn x => f x)
fun happ'(e1, e2) = 
  let val e1' = e1
  val e2' = e2
  in (case e1' of HAbs(f) => f e2' 
                              | _  => HApp(e1', e2'))
  end
fun hsym'(s) = HSym(s)
\end{verbatim}
Deforestation (2)

Simplify by η-reduction & inlining

```ml
val habs' = HAbs
fun happ'(HAbs(f), e2) = f e2
  | happ'(e1, e2) = HApp(e1, e2)
val hsym' = HSym

- conv
  (habs'(fn x => happ'(habs'(fn y => y), x)));
> val it = Abs ("x1",Var "x1") : exp
```
Comparison

Online TDPE $\approx$ Cogen approach to online SDPE

Reification operator
  $\approx$ Converter from HOAS to FOAS

Value destructors extended for residual code
  $\approx$ HOAS constructors composed with PE

They are more similar in dynamically-typed languages (e.g. Scheme) than in statically-typed ones (e.g. ML)
[Sumii & Kobayashi 99]
Related Work

• [Helsen & Thiemann 98]
  Pointed out similarity between offline TDPE and cogen approach to offline SDPE
  (C.f. Our PE is online.)

• [Sheard 97]
  Extended Danvy's TDPE in various but *ad hoc* ways such as lazy reflection, type passing, etc.
  (C.f. Our TDPE is more simple, efficient, and powerful.)
Conclusion

• We have:
  – Reviewed Danvy's TDPE
  – Extended it with online value destructors
  – Seen the similarity of our online TDPE and cogen approach to online SDPE

• Our future work includes:
  – More integration of SDPE and TDPE
  – More sophisticated treatment of side-effects (including non-termination)